
	Other comments:
	


1 Correction to SRVCC test case 13.4.3.1
1.1 f_UTRAN_SRVCC_WithHandover
	Object name 
	f_UTRAN_SRVCC_WithHandover

	Reason for change
	1. Since PS Domain is not handed over to UTRAN, when RAU request is received it would be a new registration procedure for PS and therefore should be received as a Initial Direct transfer message.
2. TMSI reallocation message should be sent as a CS-domain message (not applicable for PS)

	Summary of change
	1. Changed the parameter to true in a_UTRAN_ReceiveRauRequest
2. Used cas_DataReq_CS for TMSI reallocation message



	Source of change
	

	Module
	/UTRAN/UTRAN_CommonProcedures.ttcn


Before:

	...
function f_UTRAN_SRVCC_WithHandover(UTRAN_CellId_Type p_CellId,
                                      template (omit) PDP_ContextStatus p_PDP_ContextStatus := omit) runs on UTRAN_PTC // @sic R5-133604 sic@
  {
    var RRC_DATA_IND v_RRC_DATA_IND;
    var B3_Type v_UpdateType;
    var default v_MyDefaultVar := null;
    var DRX_CycleLengthStructure v_DRX_CycleLengthStructure := f_UTRAN_CellInfo_GetDRX_CycleLength(p_CellId);
    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();
    var PLMN_Identity v_PLMN_Identity := f_UTRAN_CellInfo_GetPLMNIdentity(p_CellId);
    var octetstring v_PLMN := f_U_Asn2Nas_PlmnId(v_PLMN_Identity);
    var octetstring v_LAC := f_UTRAN_CellInfo_GetLAC(p_CellId);
    var octetstring v_RAC := f_UTRAN_CellInfo_GetRAC(p_CellId);
    var octetstring v_PSSpecificNAS_Info := v_RAC & bit2oct(f_UTRAN_CellInfo_GetNMO(p_CellId));
    var octetstring v_CSSpecificNAS_Info := f_UTRAN_CellInfo_GetT3212(p_CellId) & int2oct(f_UTRAN_CellInfo_GetAttFlag(p_CellId), 1);
    var template (omit) PDP_ContextStatus v_PDP_ContextStatus;
    //Step 1: Receive HANDOVER TO UTRAN COMPLETE
    U_AM.receive (car_RLC_Data_Ind(utran_CellDedicated, // @sic R5s130561 sic@
                                   tsc_RB2,
                                   cr_HandoverToUTRANComplete));
    //Activate additional default behaviour to capture the Routing Area Update Request message
    v_MyDefaultVar := activate(a_UTRAN_ReceiveRauRequest(true));    
    //Steps 2-3
    f_UTRAN_ConfigureIntegrityAfterHO(p_CellId, cs_domain);
    v_SecurityInfo := f_UTRAN_Security_Get();
    //Step 4-5
    f_UTRAN_MobilityInformation(p_CellId, cs_CNInfo_PSCS(v_PLMN_Identity,
                                                         v_LAC,
                                                         v_DRX_CycleLengthStructure,
                                                         v_PSSpecificNAS_Info,
                                                         v_CSSpecificNAS_Info));
    //Step 6
    U_Dc.send (cas_DataReq_PS(utran_CellDedicated, tsc_RB3, cs_TMSI_ReallocCmd(v_PLMN, v_LAC, cs_MobileIdTMSIorPTMSI(omit, px_TMSI_Def))));    
    //Step 7
    //Receive TMSI Reallocation Complete
    // ToDo, need to confirm if uplink direct transfer needs to be allowed.
    alt {
      [] U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_TMSI_ReallocCmpl))
        {}
      [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_TMSI_ReallocCmpl))
        {}
    }
...


After:

	...

function f_UTRAN_SRVCC_WithHandover(UTRAN_CellId_Type p_CellId,
                                      template (omit) PDP_ContextStatus p_PDP_ContextStatus := omit) runs on UTRAN_PTC // @sic R5-133604 sic@
  {
    var RRC_DATA_IND v_RRC_DATA_IND;
    var B3_Type v_UpdateType;
    var default v_MyDefaultVar := null;
    var DRX_CycleLengthStructure v_DRX_CycleLengthStructure := f_UTRAN_CellInfo_GetDRX_CycleLength(p_CellId);
    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();
    var PLMN_Identity v_PLMN_Identity := f_UTRAN_CellInfo_GetPLMNIdentity(p_CellId);
    var octetstring v_PLMN := f_U_Asn2Nas_PlmnId(v_PLMN_Identity);
    var octetstring v_LAC := f_UTRAN_CellInfo_GetLAC(p_CellId);
    var octetstring v_RAC := f_UTRAN_CellInfo_GetRAC(p_CellId);
    var octetstring v_PSSpecificNAS_Info := v_RAC & bit2oct(f_UTRAN_CellInfo_GetNMO(p_CellId));
    var octetstring v_CSSpecificNAS_Info := f_UTRAN_CellInfo_GetT3212(p_CellId) & int2oct(f_UTRAN_CellInfo_GetAttFlag(p_CellId), 1);
    var template (omit) PDP_ContextStatus v_PDP_ContextStatus;
    //Step 1: Receive HANDOVER TO UTRAN COMPLETE
    U_AM.receive (car_RLC_Data_Ind(utran_CellDedicated, // @sic R5s130561 sic@
                                   tsc_RB2,
                                   cr_HandoverToUTRANComplete));
    //Activate additional default behaviour to capture the Routing Area Update Request message
    v_MyDefaultVar := activate(a_UTRAN_ReceiveRauRequest(false)); //SRVCC 13.4.3.1 Routing Area update request is received as an initial direct transfer message
    //Steps 2-3
    f_UTRAN_ConfigureIntegrityAfterHO(p_CellId, cs_domain);
    v_SecurityInfo := f_UTRAN_Security_Get();
    //Step 4-5
    f_UTRAN_MobilityInformation(p_CellId, cs_CNInfo_PSCS(v_PLMN_Identity,
                                                         v_LAC,
                                                         v_DRX_CycleLengthStructure,
                                                         v_PSSpecificNAS_Info,
                                                         v_CSSpecificNAS_Info));
    //Step 6
    U_Dc.send (cas_DataReq_CS(utran_CellDedicated, tsc_RB3, cs_TMSI_ReallocCmd(v_PLMN, v_LAC, cs_MobileIdTMSIorPTMSI(omit, px_TMSI_Def)))); //SRVCC 13.4.3.1 this message is a CS domain message
    //Step 7
    //Receive TMSI Reallocation Complete
    // ToDo, need to confirm if uplink direct transfer needs to be allowed.
    alt {
      [] U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_TMSI_ReallocCmpl))
        {}
      [] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_TMSI_ReallocCmpl))
        {}
    }
...


1.2 cs_TMSI_ReallocCmplCmd_Common
	Object name 
	cs_TMSI_ReallocCmplCmd_Common

	Reason for change
	The message type coded in TMSI reallocation message is incorrect.According to 24.008 sec 10.4 it should be 00011010

	Summary of change
	

	Source of change
	

	Module
	/Common/IRAT/NAS_24008Templates_IRAT.ttcn


Before:

	...
  template (value) TMSIREALLOCATIONCOMMAND cs_TMSI_ReallocCmplCmd_Common(octetstring p_PLMN,
                                                                           octetstring p_LAC,
                                                                           template (omit) MobileIdentity p_MobileId) :=
  { /* @status    APPROVED (LTE_IRAT) */
    skipIndicator   := '0000'B,
    mMProtocolDiscriminator := '0101'B,
    msgType         := '00001010'B,
    locAreaId       := cs_LAI(omit, p_PLMN, p_LAC),
    mobileId        := p_MobileId
  };
...


After:

	...

  template (value) TMSIREALLOCATIONCOMMAND cs_TMSI_ReallocCmplCmd_Common(octetstring p_PLMN,
                                                                           octetstring p_LAC,
                                                                           template (omit) MobileIdentity p_MobileId) :=
  { /* @status    APPROVED (LTE_IRAT) */
    skipIndicator   := '0000'B,
    mMProtocolDiscriminator := '0101'B,
    msgType         := '00011010'B,// SRVCC 13.4.3.1 sec 10.4 24.008 should be 11010
    locAreaId       := cs_LAI(omit, p_PLMN, p_LAC),
    mobileId        := p_MobileId
  };
...


1.3 f_UTRAN_ConfigureCipheringBeforeHO_SRVCC
	Object name 
	f_UTRAN_ConfigureCipheringBeforeHO_SRVCC

	Reason for change
	When performing Handover the CS-domain is only handed over to UTRA and therefore the keys used should be based on CS-domain parameters.

	Summary of change
	Removed “f_UTRAN_SS_DownloadSecurityKey(p_CellId, ps_domain);

	Source of change
	

	Module
	Common/UTRAN/UTRAN_CommonFunctions.ttcn


Before:

	...
function f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  {
    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();
    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);
    var template (value) RB_ActivationTimeInfoList  v_RB_ActivationTimeInfoList := cs_RB_ActTimeInfoListSRBs(0, 0, 0, 0);
    var RLC_SequenceNumber  v_RLC_SN_RB1, v_RLC_SN_RB2, v_RLC_SN_RB3, v_RLC_SN_RB4, v_RLC_SN_RAB;
    var UTRAN_CFN_Info_Type v_CfnInfo := f_CalculateActTime (p_CellId);
    if (v_SecurityInfo.start_CS == '11111111111111111111'B) {
      FatalError(__FILE__, __LINE__, "invalid CS Start Value");
    }
    if (v_SecurityInfo.start_PS == '11111111111111111111'B) {
      FatalError(__FILE__, __LINE__, "invalid PS Start Value");
    }
    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_SecurityInfo.start_CS);
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_SecurityInfo.start_PS);
    v_SecurityInfo.dL_CipherMode := valueof(cs_CipheringModeCmdOn(px_UTRAN_CipheringAlgorithm));
    v_SecurityInfo.cs_cipheringStarted := true;
    v_SecurityInfo.recentSecureDomain := cs_domain;
    f_UTRAN_Security_Set(v_SecurityInfo);
    f_UTRAN_SS_DownloadSecurityKey(p_CellId, cs_domain);
    f_UTRAN_SS_DownloadSecurityKey(p_CellId, ps_domain);
    v_RLC_SN_RB1  := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB1);
    v_RLC_SN_RB2  := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB2);
    v_RLC_SN_RB3  := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB3);
    v_RLC_SN_RB4  := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB4);
    f_UTRAN_ActivateCiphering_RLC_DL_SS_RB_1_2_3_4(p_CellId, v_SecurityInfo.recentSecureDomain, v_SecurityInfo.dL_CipherMode, v_RLC_SN_RB1, v_RLC_SN_RB2, v_RLC_SN_RB3, v_RLC_SN_RB4);
    // UL Ciphering for SRB's using CS domain
    v_RB_ActivationTimeInfoList := cs_RB_ActTimeInfoListSRBs(0,0,0,0);
    f_UTRAN_ActivateCiphering_RLC_UL(f_UTRAN_CellInfo_GetFDD_TDD(p_CellId), cs_domain, v_RB_ActivationTimeInfoList);
    // Start Ciphering in MAC for CS TM bearers
    f_UTRAN_CMAC_SecurityMode_Config(cs_domain,
                                     v_SecurityInfo.start_CS,
                                     v_SecurityInfo.authKeys.AuthCS.CKey);
    f_UTRAN_CMAC_UL_DL_CipherCfg(p_CellId,
                                 v_SecurityInfo.dL_CipherMode,
                                 v_CfnInfo.actTime ,
                                 notInc);
    // PS ciphering if PS bearers are configured.
    select (v_CellConfig) {
      case (cell_Four_DTCH_PS_CS) {
        v_RLC_SN_RAB := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB20);
        v_RB_ActivationTimeInfoList := cs_RB_ActTimeInfoList20(v_RLC_SN_RAB);
        f_UTRAN_CRLC_DL_CipherCfgRB(p_CellId, ps_domain, v_SecurityInfo.dL_CipherMode, v_CellConfig, false); //@sic R5s120113 sic@
        f_UTRAN_ActivateCiphering_RLC_UL(f_IRAT_GetUTRAN_FDD_TDD_Mode(), ps_domain, v_RB_ActivationTimeInfoList);
      }
    }
  } 
...


After:

	...
function f_UTRAN_ConfigureCipheringBeforeHO_SRVCC(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  {
    var UTRAN_SecurityInfo_Type v_SecurityInfo := f_UTRAN_Security_Get();
    var RB_ConfigType v_CellConfig := f_UTRAN_CellInfo_GetConfigType(p_CellId);
    var template (value) RB_ActivationTimeInfoList  v_RB_ActivationTimeInfoList := cs_RB_ActTimeInfoListSRBs(0, 0, 0, 0);
    var RLC_SequenceNumber  v_RLC_SN_RB1, v_RLC_SN_RB2, v_RLC_SN_RB3, v_RLC_SN_RB4, v_RLC_SN_RAB;
    var UTRAN_CFN_Info_Type v_CfnInfo := f_CalculateActTime (p_CellId);
    if (v_SecurityInfo.start_CS == '11111111111111111111'B) {
      FatalError(__FILE__, __LINE__, "invalid CS Start Value");
    }
    if (v_SecurityInfo.start_PS == '11111111111111111111'B) {
      FatalError(__FILE__, __LINE__, "invalid PS Start Value");
    }
    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_SecurityInfo.start_CS);
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_SecurityInfo.start_PS);
    v_SecurityInfo.dL_CipherMode := valueof(cs_CipheringModeCmdOn(px_UTRAN_CipheringAlgorithm));
    v_SecurityInfo.cs_cipheringStarted := true;
    v_SecurityInfo.recentSecureDomain := cs_domain;
    f_UTRAN_Security_Set(v_SecurityInfo);
    f_UTRAN_SS_DownloadSecurityKey(p_CellId, cs_domain);
    //f_UTRAN_SS_DownloadSecurityKey(p_CellId, ps_domain);    
    v_RLC_SN_RB1  := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB1);
    v_RLC_SN_RB2  := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB2);
    v_RLC_SN_RB3  := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB3);
    v_RLC_SN_RB4  := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB4);
    f_UTRAN_ActivateCiphering_RLC_DL_SS_RB_1_2_3_4(p_CellId, v_SecurityInfo.recentSecureDomain, v_SecurityInfo.dL_CipherMode, v_RLC_SN_RB1, v_RLC_SN_RB2, v_RLC_SN_RB3, v_RLC_SN_RB4);
    // UL Ciphering for SRB's using CS domain
    v_RB_ActivationTimeInfoList := cs_RB_ActTimeInfoListSRBs(0,0,0,0);
    f_UTRAN_ActivateCiphering_RLC_UL(f_UTRAN_CellInfo_GetFDD_TDD(p_CellId), cs_domain, v_RB_ActivationTimeInfoList);
    // Start Ciphering in MAC for CS TM bearers
    f_UTRAN_CMAC_SecurityMode_Config(cs_domain,
                                     v_SecurityInfo.start_CS,
                                     v_SecurityInfo.authKeys.AuthCS.CKey);
    f_UTRAN_CMAC_UL_DL_CipherCfg(p_CellId,
                                 v_SecurityInfo.dL_CipherMode,
                                 v_CfnInfo.actTime ,
                                 notInc);
    // PS ciphering if PS bearers are configured.
    select (v_CellConfig) {
      case (cell_Four_DTCH_PS_CS) {
        v_RLC_SN_RAB := f_UTRAN_CRLC_SequenceNumber_GetCount_C_LSB_DL(tsc_RB20);
        v_RB_ActivationTimeInfoList := cs_RB_ActTimeInfoList20(v_RLC_SN_RAB);
        f_UTRAN_CRLC_DL_CipherCfgRB(p_CellId, ps_domain, v_SecurityInfo.dL_CipherMode, v_CellConfig, false); //@sic R5s120113 sic@
        f_UTRAN_ActivateCiphering_RLC_UL(f_IRAT_GetUTRAN_FDD_TDD_Mode(), ps_domain, v_RB_ActivationTimeInfoList);
      }
    }
  }
...


1.4 fl_TC_13_4_3_1_Body
	Object name 
	fl_TC_13_4_3_1_Body

	Reason for change
	1. At Step 1, when you send a trigger to UTRA, you should not be waiting for a trigger back from UTRAN, as on the UTRAN side, a trigger is not sent back.UTRAN PTC changes the power level (T0) and waits for another trigger for EUTRA PTC

2. Measurement control configured is configuring RSCP reporting values for UTRAN, but the receive template is expecting EcNo values.

	Summary of change
	1. Changed to use f_IRAT_SendCoOrd( UTRAN, cms_IRAT_Trigger );
2. Changed to use cr_MeasurementReport_Utran_1Entry_FDD_RSCP

	Source of change
	

	Module
	


Before:

	...
   function fl_TC_13_4_3_1_Body() runs on EUTRA_PTC
  {
    var AbsoluteCellPower_Type v_T1_RS_EPRE_Cell1 := -84;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NasCount_Type  v_NasCountDL;
    var integer v_UTRAN_PhysChId_Cell5;
    var octetstring v_HandoverToUtranCommand, v_Nas_SecurityParamFromEUTRA;
    var Frequency_IE_Type   v_Frequency_IE_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs (eutra_Cell1);
    var ChannelBandwidthDependency_Type  v_ChBandDependency_Cell1 := f_EUTRA_BandDependentParam(v_Frequency_IE_Cell1.DL_ChBandwidth, v_Frequency_IE_Cell1.UL_ChBandwidth);
    var integer v_UTRAN_ARFCN_Cell5;
    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;
    var CoOrd_UTRANSysInfoFDD_Type v_UTRANSysInfoFDD;
    var CoOrd_UTRANSysInfoTDD_Type v_UTRANSysInfoTDD;
    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);
    if (ischosen (v_UTRANSysInfo.Utran[0].FDD)) {
      v_UTRANSysInfoFDD := v_UTRANSysInfo.Utran[0].FDD;
      v_UTRAN_ARFCN_Cell5 := v_UTRANSysInfo.Utran[0].FDD.Freq;
      v_UTRAN_PhysChId_Cell5 := v_UTRANSysInfo.Utran[0].FDD.ScramblingCode;
      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_Speech_FDD(v_UTRANSysInfoFDD.UplinkScrambingCode,
                                                                       v_UTRANSysInfoFDD.ScramblingCode,
                                                                       v_UTRANSysInfoFDD.Freq);
    }
    else {
      v_UTRANSysInfoTDD := v_UTRANSysInfo.Utran[0].TDD;
      v_UTRAN_ARFCN_Cell5 := v_UTRANSysInfo.Utran[0].TDD.Freq;
      v_UTRAN_PhysChId_Cell5 := v_UTRANSysInfo.Utran[0].TDD.CellParametersID;
      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_Speech_TDD(v_UTRANSysInfoTDD.CellParametersID,
                                                                       v_UTRANSysInfoTDD.Freq);
    }
    // @siclog "Step 1" siclog@
    // The SS configures UTRA cell 5 to reference configuration according 36.508 table 4.8.3-1, condition UTRA PS RB + Speech .
    f_EUTRA_SendTriggerToOtherRAT(true);
    // @siclog "Steps 2-27" siclog@
    // Make the UE attempt an IMS speech call
    f_EUTRA_GenericIMS_MT_SpeechCall(eutra_Cell1);
    // @siclog "Step 28" siclog@
    // The SS transmits an RRCConnectionReconfiguration message on Cell 1 to setup inter RAT measurement and reporting for event B2.
    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                  cs_TimingInfo_Now,
                                  cs_RRCConnectionReconfiguration_SRVCC_FDD(v_RRC_TI,
                                                                             v_Frequency_IE_Cell1.UL_DL_Earfcn.dl_CarrierFreq,
                                                                             v_ChBandDependency_Cell1.AllowedMeasBandwidth,
                                                                             v_UTRAN_ARFCN_Cell5,
                                                                             v_UTRAN_PhysChId_Cell5)));
    // @siclog "Step 29" siclog@
    // The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));
    // @siclog "Step 30" siclog@
    // The SS changes the power level for Cell 1 and Cell 5 according to the row "T1" in table 13.4.3.2.3.2-1
    f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger);
    f_EUTRA_SetCellPower(eutra_Cell1, v_T1_RS_EPRE_Cell1);
    // @siclog "Step 31" siclog@
    // The UE transmits a MeasurementReport message on Cell 1 to report event B2 for Cell 5.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_Utran_1Entry_FDD(1, v_UTRAN_PhysChId_Cell5)));
    // @siclog "Step 31A - 31B" siclog@
    f_EUTRA_UTRA_Capability(eutra_Cell1, true); //@sic R5-120776 sic@
    f_IRAT_WaitForCoOrd_Trigger(UTRAN);
    // nas-securityParamFromEUTRA is the 4 least significant bits of the NAS downlink COUNT value according to Table 13.1.4.3.3-6
    v_NasCountDL := f_EUTRA_SS_NAS_CountGetDLNext();
    v_Nas_SecurityParamFromEUTRA := substr(v_NasCountDL, 3, 1);
    // @siclog "Step 32" siclog@
    // The SS transmits a MobilityFromEUTRACommand message on Cell 1.
    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                 cs_TimingInfo_Now,
                                 cs_MobilityFromEutraCommand_Handover(tsc_RRC_TI_Def, true, utra, v_HandoverToUtranCommand, v_Nas_SecurityParamFromEUTRA, omit)));
    // Release DRX configuration
    f_EUTRA_SS_DrxCtrlConfig(eutra_Cell1, cs_DrxCtrl_None);
    // Release SRBs and DRBs
    f_EUTRA_SS_SRBs_DRBs_Release(eutra_Cell1, cs_TimingInfo_Now);
  }
...

	


After:

	...
    function fl_TC_13_4_3_1_Body() runs on EUTRA_PTC
  {
    var AbsoluteCellPower_Type v_T1_RS_EPRE_Cell1 := -84;
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;
    var NasCount_Type  v_NasCountDL;
    var integer v_UTRAN_PhysChId_Cell5;
    var octetstring v_HandoverToUtranCommand, v_Nas_SecurityParamFromEUTRA;
    var Frequency_IE_Type   v_Frequency_IE_Cell1 := f_EUTRA_CellInfo_GetFrequencyIEs (eutra_Cell1);
    var ChannelBandwidthDependency_Type  v_ChBandDependency_Cell1 := f_EUTRA_BandDependentParam(v_Frequency_IE_Cell1.DL_ChBandwidth, v_Frequency_IE_Cell1.UL_ChBandwidth);
    var integer v_UTRAN_ARFCN_Cell5;
    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;
    var CoOrd_UTRANSysInfoFDD_Type v_UTRANSysInfoFDD;
    var CoOrd_UTRANSysInfoTDD_Type v_UTRANSysInfoTDD;
    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);
    if (ischosen (v_UTRANSysInfo.Utran[0].FDD)) {
      v_UTRANSysInfoFDD := v_UTRANSysInfo.Utran[0].FDD;
      v_UTRAN_ARFCN_Cell5 := v_UTRANSysInfo.Utran[0].FDD.Freq;
      v_UTRAN_PhysChId_Cell5 := v_UTRANSysInfo.Utran[0].FDD.ScramblingCode;
      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_Speech_FDD(v_UTRANSysInfoFDD.UplinkScrambingCode,
                                                                       v_UTRANSysInfoFDD.ScramblingCode,
                                                                       v_UTRANSysInfoFDD.Freq);
    }
    else {
      v_UTRANSysInfoTDD := v_UTRANSysInfo.Utran[0].TDD;
      v_UTRAN_ARFCN_Cell5 := v_UTRANSysInfo.Utran[0].TDD.Freq;
      v_UTRAN_PhysChId_Cell5 := v_UTRANSysInfo.Utran[0].TDD.CellParametersID;
      v_HandoverToUtranCommand := f_U_EncodeHandoverToUtran_Speech_TDD(v_UTRANSysInfoTDD.CellParametersID,
                                                                       v_UTRANSysInfoTDD.Freq);
    }
    // @siclog "Step 1" siclog@
    // The SS configures UTRA cell 5 to reference configuration according 36.508 table 4.8.3-1, condition UTRA PS RB + Speech .
    f_IRAT_SendCoOrd( UTRAN, cms_IRAT_Trigger ); // Send trigger
    // @siclog "Steps 2-27" siclog@
    // Make the UE attempt an IMS speech call
    f_EUTRA_GenericIMS_MT_SpeechCall(eutra_Cell1);
    // @siclog "Step 28" siclog@
    // The SS transmits an RRCConnectionReconfiguration message on Cell 1 to setup inter RAT measurement and reporting for event B2.
    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                  cs_TimingInfo_Now,
                                  cs_RRCConnectionReconfiguration_SRVCC_FDD(v_RRC_TI,
                                                                             v_Frequency_IE_Cell1.UL_DL_Earfcn.dl_CarrierFreq,
                                                                             v_ChBandDependency_Cell1.AllowedMeasBandwidth,
                                                                             v_UTRAN_ARFCN_Cell5,
                                                                             v_UTRAN_PhysChId_Cell5)));
    // @siclog "Step 29" siclog@
    // The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1.
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));
    // @siclog "Step 30" siclog@
    // The SS changes the power level for Cell 1 and Cell 5 according to the row "T1" in table 13.4.3.2.3.2-1
    f_IRAT_SendCoOrd(UTRAN, cms_IRAT_Trigger);
    f_EUTRA_SetCellPower(eutra_Cell1, v_T1_RS_EPRE_Cell1);
    // @siclog "Step 31" siclog@
    // The UE transmits a MeasurementReport message on Cell 1 to report event B2 for Cell 5.
    // BZ 47781
    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_MeasurementReport_Utran_1Entry_FDD_RSCP( 1, v_UTRAN_PhysChId_Cell5 ) ) ); 
    // @siclog "Step 31A - 31B" siclog@
    f_EUTRA_UTRA_Capability(eutra_Cell1, true); //@sic R5-120776 sic@
    f_IRAT_WaitForCoOrd_Trigger(UTRAN);
    // nas-securityParamFromEUTRA is the 4 least significant bits of the NAS downlink COUNT value according to Table 13.1.4.3.3-6
    v_NasCountDL := f_EUTRA_SS_NAS_CountGetDLNext();
    v_Nas_SecurityParamFromEUTRA := substr(v_NasCountDL, 3, 1);
    // @siclog "Step 32" siclog@
    // The SS transmits a MobilityFromEUTRACommand message on Cell 1.
    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                 cs_TimingInfo_Now,
                                 cs_MobilityFromEutraCommand_Handover(tsc_RRC_TI_Def, true, utra, v_HandoverToUtranCommand, v_Nas_SecurityParamFromEUTRA, omit)));
    // Release DRX configuration
    f_EUTRA_SS_DrxCtrlConfig(eutra_Cell1, cs_DrxCtrl_None);
    // Release SRBs and DRBs
    f_EUTRA_SS_SRBs_DRBs_Release(eutra_Cell1, cs_TimingInfo_Now);
  }
  /*
...


1.5 f_EUTRA_GenericIMS_MT_SpeechCall
	Object name 
	f_EUTRA_GenericIMS_MT_SpeechCall

	Reason for change
	1. When a VoIP call is established, the UE starts sending RTP packets on the voice media port at a rate of one voice packet every 20 mseconds. The current IMS test model does not support RTP packet handling and the test case would stop due to unexpected data received.
2. Activate EPS Bearer Context Request is sent twice. 

3. Receive statement for Activate EPS Bearer Context Accept is missing.

	Summary of change
	1. Similar to change 3 proposed in R5s130681, the RTP port configuration was added to enable the SS omitting RTP packets received.

2. Second sent of Activate EPS Bearer Context EPS Request was commented.

3. Receive for Activate EPS Bearer Context Accept was added.

	Source of change
	

	Module
	/EUTRA/EUTRA_CommonSRVCC.ttcn


Before:

	...
   function f_EUTRA_GenericIMS_MT_SpeechCall(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  {
    var ProcedureTransactionIdentifier v_EpsProcedureTransactionId := '00'O;
    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated;
    v_RadioResourceConfigDedicated := cs_RadioResourceConfig_DRB_RECONFIG(cs_DRB_ToAddMod_DEFAULT_UM_RoHC(tsc_DRB2),
                                                                          cs_508_MAC_MainConfig_Explicit_RBC_DrxS);
    //Steps 1-8
    f_EUTRA_RbEst_Def(p_CellId);
    // Send trigger to IMS PTC
    // Steps 9 to 12 are executed in IMS PTC
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);
    //Wait Trigger from IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);
    //Steps 13-15
    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_Cell1,
                                    cs_TimingInfo_Now,
                                    cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,
                                                                           omit,
                                                                           omit,
                                                                           v_RadioResourceConfigDedicated ,
                                                                           omit),
                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                   cs_508_ActDedicatedEPSBearerCxtReq(tsc_EpsDedicatedBearerId,
                                                                                      v_EpsProcedureTransactionId,
                                                                                      cs_LinkedId(hex2bit(tsc_EpsDefaultBearerId)),
                                                                                      cds_EPS_QoS_Dedicated_4, // According to Context#4 in TS 36.508
                                                                                      f_EUTRA_Get_508_TrafficFlowTemplate(4))))); // According to Context#4 in TS 36.508
    //Step 2
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS_DrbReconfig(p_CellId,
                                                                 tsc_RRC_TI_Def,
                                                                 {tsc_DRB2},
                                                                 {cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                                                 cs_508_ActDedicatedEPSBearerCxtReq(tsc_EpsDedicatedBearerId,
                                                                                                                    tsc_PTI_Unassigned,
                                                                                                                    cs_LinkedId(hex2bit(tsc_EpsDefaultBearerId)),
                                                                                                                    cs_508_EPS_QoS_Dedicated_1,
                                                                                                                    f_EUTRA_Get_508_TrafficFlowTemplate(1, tsc_EpsDedicatedBearerId)))},
                                                                 UM_Mode);
    // Send trigger to IMS PTC
    // Steps 16 to 26 are executed in IMS PTC
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);
    //Wait Trigger from IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);
  }
...


After:

	...
  function f_EUTRA_GenericIMS_MT_SpeechCall(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
  {
    var ProcedureTransactionIdentifier v_EpsProcedureTransactionId := '00'O;
    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated;
    v_RadioResourceConfigDedicated := cs_RadioResourceConfig_DRB_RECONFIG(cs_DRB_ToAddMod_DEFAULT_UM_RoHC(tsc_DRB2),
                                                                          cs_508_MAC_MainConfig_Explicit_RBC_DrxS);
    //Steps 1-8
    f_EUTRA_RbEst_Def(p_CellId);
    // Send trigger to IMS PTC
    // Steps 9 to 12 are executed in IMS PTC
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);
    //Wait Trigger from IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);
   // Configure RTP port for VoIP data
   if (pc_IPv4) {
   f_IP_ConfigureRTP(IP, p_CellId, PDN_1, 2, px_IPv4_Address1_NW, 60350);  // RTP port is 60350
    } 
    if (pc_IPv6) {
   f_IP_ConfigureRTP(IP, p_CellId, PDN_1, 2, px_IPv6_Address1_NW, 60350);  
    } 
    //Steps 13-15
    SRB.send(cas_SRB1_RrcNasPdu_REQ(eutra_Cell1,
                                    cs_TimingInfo_Now,
                                    cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def,
                                                                           omit,
                                                                           omit,
                                                                           v_RadioResourceConfigDedicated ,
                                                                           omit),
                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                   cs_508_ActDedicatedEPSBearerCxtReq(tsc_EpsDedicatedBearerId,
                                                                                      v_EpsProcedureTransactionId,
                                                                                      cs_LinkedId(hex2bit(tsc_EpsDefaultBearerId)),
                                                                                      cds_EPS_QoS_Dedicated_4, // According to Context#4 in TS 36.508
                                                                                      f_EUTRA_Get_508_TrafficFlowTemplate(4))))); // According to Context#4 in TS 36.508
    //Step 2
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message?
    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,
                                    cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
     //BZ 47781 - Message sent twice                              
    /*f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS_DrbReconfig(p_CellId,
                                                                 tsc_RRC_TI_Def,
                                                                 {tsc_DRB2},
                                                                 {cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                                                 cs_508_ActDedicatedEPSBearerCxtReq(tsc_EpsDedicatedBearerId,
                                                                                                                    tsc_PTI_Unassigned,
                                                                                                                    cs_LinkedId(hex2bit(tsc_EpsDefaultBearerId)),
                                                                                                                    cs_508_EPS_QoS_Dedicated_1,
                                                                                                                    f_EUTRA_Get_508_TrafficFlowTemplate(1, tsc_EpsDedicatedBearerId)))},
                                                                 UM_Mode);*/
    //BZ 47781 - Receive missing                                                            
    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                                         cr_508_ActDedicatedEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));                                                             
    // Send trigger to IMS PTC
    // Steps 16 to 26 are executed in IMS PTC
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);
    //Wait Trigger from IMS PTC
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);
  }
...


